
Femapモデルファイルの読込み例 Word マクロ使用説明書 

 

 

 

 

  

ソフトの概要 

本資料は Femap API を使用した例として NASTRAN 板及び梁要素モデルの固有値

解析結果を使用し、起動している Femap のモデルファイルの内容を Word 文書に出

力した説明書です。 

APIとはアプリケーション プログラミング インターフェイスことです。Visual 

Basic、VBA(Excel, Word, Access, ... )、C、C++、さらには Femap Basic Scripting 

Language から呼び出し可能な関数群です。 

 

出力例 

①｢物性値一覧表｣ ：物性値 

②｢板部材特性一覧表｣ ：板要素に使用する板厚 

③｢梁部材特性一覧表｣ ：梁要素に使用する部材特性 

④｢質量一覧表｣ ：節点質量 

⑤｢固有振動数一覧表｣ ：固有値計算結果の固有振動数 

 

注意事項 

①本 VBAは個人所有ですが、参考にして改編を行ってもらって構いません。 

②テストモデルでの機能のチェックは行っていますが、処理できないデータもあ

ると思います。また、エラー処理等は十分に行っていません。 

使用者の責任において使用して下さい。 

③不具合の修正、機能の追加については使用者が行って下さい。 

④本 VBAの内容についての質問は受け付けません。ただし、費用が発生しても問

題なければ質問、機能の追加または作成については受け付けます。 

 

  開発者:茨木 栄 Mail:sakae-ibaraki@mbr.nifty.com 

改正来歴 

システム

Ver. 
改正日 改正内容 

0.0 0.1 2021/11/26 初版 

添付ファイル 

① 使用説明書 

ZW-FmpGetMdl_Ver.0.1_Femapモデルファイルの読込み例 Word マクロ 

使用説明書.pdf 

②VBAファイル 

ZW-FmpGetMdl_Ver.0.1.docm 

③テストファイル 

・Femapモデルファイル 

\TestData\METH.modfem 

・NASTRANインプットデータ 

\TestData\METH000.dat 

・NASTRANアウトプットリスト 

\TestData\meth000.f06 

1.概要 

本資料は、Word VBAを使用し起動している Femapのモデルファイルを Femap APIを使用して

読み込み、Word文書に出力した例をまとめたものです。 

なお、モデルのデータ入出力以外に変形図、コンタ図等の図形の表示、ファイルへの保存、

クリップボートへのコピー等も行えます。 

API の詳細については Femap がインストールされているフォルダ内の以下のファイルを参照

して下さい。日本語版は翻訳したものであるため英語版も参考にして下さい。また、参考ブロ

グラムもインストールされているフォルダ内の以下のフォルダに複数保存されています。 

 

①日本語版説明書 

インストールフォルダ\pdf\api.pdf 

②英語版説明書 

インストールフォルダ\pdf\EN\api.pdf 

③参考プログラム 

インストールフォルダ\API 

 

注:本資料では起動している Femapのモデルファイルを処理していますが、ファイル名を指定

して Femapを起動し処理することも可能です。 

2.使用環境 

使用した Word VBAのバージョンと Femapのバージョンを以下に示す。 

 

 

図 2-1 Word VBA バージョン 

 

 

図 2-2 Femap バージョン 

3.使用例 

VBAが保存されている｢ZW-FmpGetMdl_Ver.0.1.docm｣を開くと図 3-1が表示される。 

対象とする Femap モデルファイルを表示し、各ボタンをクリックして Femap のモデルデータ

を読み込み、読込み結果をワード文書に出力する。出力結果の例を 4項の表 4-1-1～表 4-5-1に

示す。 

なお、表 4-1-1～表 4-5-1に示す一覧表の内容は添付ファイルの｢\TestData\METH.modfem｣を

Femapで表示して実行した結果です。 

 

 

図 3-1 Word画面 

4.検証 

表 4-1-1～表 4-5-2に Femapモデルファイルから読み込んだ Wordの結果と NASTRANのインプ

ット及びアウトプットリストを示す。表 4-4-1 及び表 4-4-2 は桁数が異なるがそれ以外につい

ては同じ値となっている。 

なお、表 4-1-1～表 4-5-1は添付ファイルの｢\TestData\METH.modfem｣を Femapで表示して実

行した結果です。 

 

 

表 4-1-1 物性値一覧表 

物性値一覧表 

MID タイトル 
ヤング率 
(N/mm2) 

ポアソン比 
密度 

(N/mm3) 

2 等方性 ﾏﾃﾘｱﾙ 2.0000×105 0.3 7.8500×10-9 

3 等方性 ﾏﾃﾘｱﾙ 3.0000×105 0.3 7.8500×10-9 

 

 

表 4-1-2 NASTRANインプットデータ｢MAT1｣ 

 

表 4-2-1 板部材特性一覧表 

板部材特性一覧表 

PID タイトル MID 
板厚 
(mm) 

31 T=10 2 10.0 

41 T=5 3 5.0 

 

 

表 4-2-2 NASTRANインプットデータ｢PSHELL｣ 

 

  

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

$ Femap with NX Nastran Material 2 : 等方性 ﾏﾃﾘｱﾙ 

MAT1           2 200000.76923.08      .3  7.85-9      0.      0.         

$ Femap with NX Nastran Material 3 : 等方性 ﾏﾃﾘｱﾙ 

MAT1           3 300000. 115385.      .3  7.85-9      0.      0.         

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

$ Femap with NX Nastran Property 31 : T=10 

PSHELL        31       2     10.       2               2              0.+        

+             0.                 

$ Femap with NX Nastran Property 41 : T=5 

PSHELL        41       3      5.       3               3              0.+        

+             0.                 

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------*    

表 4-3-1 梁部材特性一覧表 

梁要素部材特性一覧表 

PID タイトル MID 
断面積 
A(mm2) 

断面二次 
IY(mm4) 

断面二次 
IZ(mm4) 

ねじり 
J(mm4) 

4 □50X30 3 1.5000×103 3.1250×105 1.1250×105 2.8195×105 

5 ◎50X2 3 4.4296×102 1.2281×105 1.2281×105 2.4558×105 

 

 

表 4-3-2 NASTRANインプットデータ｢PBAR｣ 

 

 

表 4-4-1 質量一覧表 

質量一覧表 

PID タイトル 
質量 

(N/mm3) 
オフセット X 

(mm) 
オフセット Y 

(mm) 
オフセット Z 

(mm) 

1 質量 ﾌﾟﾛﾊﾟﾃｨ   0.015 0.0 0.0 0.0 

2 質量 ﾌﾟﾛﾊﾟﾃｨ   0.013 0.0 0.0 0.0 

 

 

表 4-4-2 NASTRANインプットデータ｢CONM2｣ 

  

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

$ Femap with NX Nastran Property 4 : □50X30 

$ Femap with NX Nastran PropShape 4 : 1,0,50.,30.,0.,0.,0.,0. 

$ Femap with NX Nastran PropOrient 4 : 1,0,0.,1.,2.,3.,4.,-1.,0.,0. 

PBAR           4       3   1500. 112500. 312500. 281948.      0.        +        

+           -15.    -25.     15.    -25.     15.     25.    -15.     25.+        

+        .850452.8499747      0. 

$ Femap with NX Nastran Property 5 : ◎50X2 

$ Femap with NX Nastran PropShape 5 : 6,0,25.,0.,0.,0.,0.,3. 

$ Femap with NX Nastran PropOrient 5 : 6,0,0.,1.,2.,3.,4.,-1.,0.,0. 

PBAR           5       3442.9646 122812. 122812. 245576.      0.        +        

+             0.    -25.     25.      0.      0.     25.    -25.      0.+        

+       .5340677.5340677      0. 

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

CONM2         65       6       0  .01502      0.      0.      0.                 

CONM2         66      12       0  .01301      0.      0.      0.                 

$------*-------*-------*-------*-------*-------*-------*-------*-------*-------* 

表 4-5-1 固有振動数一覧表 

固有振動数一覧表 

次数 
固有振動数 

(Hz) 

1 16.991 

2 25.181 

3 30.930 

4 34.506 

5 56.440 

6 70.624 

7 87.638 

8 113.123 

9 131.381 

10 148.530 

 

 

表 4-5-2 NASTRANアウトプットリスト｢固有振動数｣ 

 

  

                                              R E A L   E I G E N V A L U E S 

   MODE    EXTRACTION      EIGENVALUE            RADIANS             CYCLES            GENERALIZED         GENERALIZED 

    NO.       ORDER                                                                       MASS              STIFFNESS 

        1         1        1.139663E+04        1.067550E+02        1.699059E+01        1.000000E+00        1.139663E+04 

        2         2        2.503234E+04        1.582161E+02        2.518088E+01        1.000000E+00        2.503234E+04 

        3         3        3.776823E+04        1.943405E+02        3.093025E+01        1.000000E+00        3.776823E+04 

        4         4        4.700582E+04        2.168082E+02        3.450610E+01        1.000000E+00        4.700582E+04 

        5         5        1.257568E+05        3.546221E+02        5.643985E+01        1.000000E+00        1.257568E+05 

        6         6        1.969073E+05        4.437424E+02        7.062379E+01        1.000000E+00        1.969073E+05 

        7         7        3.032082E+05        5.506434E+02        8.763762E+01        1.000000E+00        3.032082E+05 

        8         8        5.051962E+05        7.107715E+02        1.131228E+02        1.000000E+00        5.051962E+05 

        9         9        6.814378E+05        8.254924E+02        1.313812E+02        1.000000E+00        6.814378E+05 

       10        10        8.709416E+05        9.332425E+02        1.485302E+02        1.000000E+00        8.709416E+05 

5.まとめ 

本使用例では Femapモデルファイルの読み込みしか行っていないが書き込みも可能です。 

読み込みについては必要なデータのみ処理すればよいが書き込みについては必要なデータはす

べて書き込む必要がある。なお、解析モデルをジオメトリから作成することも可能です。 

どのデータが必要かについては説明書には記載がないため、一度モデルデータを読み込んで

データの内容を確認する必要がある。 

使用した感じでは、Visual Studio等の実行ファイルを使用した方が良い。 

Femap の｢ツール｣→｢プログラミング｣→｢API プログラミング｣メニューで使用できる Visual 

Basicもあるがエディタが非常に使いにくい。 

6.Wrod VBAリスト 

Word VBAで Femap APIを使用するには VBAエディタの｢ツール｣→｢参照設定｣メニューをクリ

ックし図 6-1に示す｢参照設定｣ダイアログを表示し Femapのライブラリを参照する必要がある。 

表 6-1 に本 VBA の関数一覧表を、表 6-2～表 6-19 に本資料で使用した VBA のリストを示す。 

 

 

図 6-1 ｢参照設定｣ダイアログ 

表 6-1 関数一覧表 

№ Function Name 内容 

1  定数 

2  イベントプロシージャ 

3 FeMaterial Femapモデルファイルの MATERIALデータの一覧表出力 

4 FePropatyShell Femapモデルファイルの PROPATY(板)データの一覧表出力 

5 FePropatyBar Femapモデルファイルの PROPATY(梁)データの一覧表出力 

6 FePropatyMass Femapモデルファイルの PROPATY(質量)データの一覧表出力 

7 FeCycle 
Femapモデルファイルの OutputSetデータから固有値計算結果

の次数と周波数の一覧表出力 

8 InitWord ワードの初期設定 

9 InitDocPage ページ設定 

10 InitDocFont フォント初期設 

11 InitDocParagraphFormat 書式の初期設定 

12 SetWidth 表の列幅の設定 

13 SetHeader 表ヘッダ書式の設定 

14 SetAlignWidth 表の列の文字位置の設定 

15 Superscript 上付き、下付き文字 

16 StrFit 
数値を文字列に変換し小数部が合うように文字左にブランクを

追加 

17 SetFormat 
指数値を文字列に変換して(1.234E+02→1.234×102、 

1.234E-02→1.234×10-2)指数を上付きに設定 

18 Log10 log10 

表 6-2 定数 

 

 

  

'*********************************************************************************** 
'Femapモデルデータを読み込み一覧表の作成 

'作成 2021/08/11 茨木 栄 (jah01272@nifty.com) 
'注:Femapでモデルを表示している必要かある。 
'*********************************************************************************** 

Option Explicit 
 
'シート名 

Const TABLENAME_MAT = "物性値一覧表" 
Const TABLENAME_PSHELL = "板部材特性一覧表" 
Const TABLENAME_PBAR = "梁部材特性一覧表" 

Const TABLENAME_PMASS = "質量一覧表" 
Const TABLENAME_CYCLE = "固有振動数一覧表" 

表 6-3(1/2) イベントプロシージャ 

 

  

'｢説明書｣ボタンのクリック 
Private Sub ButHelp_Click() 
   CreateObject("Shell.Application").ShellExecute ActiveDocument.Path & "\ZW-FmpGetMdl_Ver.0.1_Femap モデルデ

ータの読込み例 Word マクロ使用説明書.pdf" 
End Sub 
 

'｢全データ｣ボタンのクリック 
Private Sub ButAll_Click() 
  Dim errMsg As String 'エラーメッセージ 

      
  System.Cursor = wdCursorWait       'マウスポインタの設定 
  Application.ScreenUpdating = False '画面の更新を停止する 

   
  '----------------------------------------------------------- 
  'ワード出力 

  '----------------------------------------------------------- 
  '新しい文書を開く 
  Documents.Add Template:="Normal", NewTemplate:=False, DocumentType:=wdNewBlankDocument 

  'ワードの初期設定 
  errMsg = InitWord 
  If Len(errMsg) <> 0 Then GoTo Last 

   
  'Femap モデルファイルの MATERIALデータを表で出力 
  errMsg = FeMaterial(True) 

  If Len(errMsg) <> 0 Then GoTo Last 
   
  'Femap モデルファイルの PROPATY(板)データを表で出力 

  errMsg = FePropatyShell(True) 
  If Len(errMsg) <> 0 Then GoTo Last 
  'Femap モデルファイルの PROPATY(梁)データを表で出力 

  errMsg = FePropatyBar(True) 
  If Len(errMsg) <> 0 Then GoTo Last 
   

  'Femap モデルファイルの PROPATY(質量)データを表で出力 
  errMsg = FePropatyMass(True) 
  If Len(errMsg) <> 0 Then GoTo Last 

   
  'Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数シートに出力 
  errMsg = FeCycle(True) 

  If Len(errMsg) <> 0 Then GoTo Last 
   
Last: 

  Application.ScreenUpdating = True '画面を更新する 
  System.Cursor = wdCursorNormal    'マウスポインタの設定 
   

  If Len(errMsg) <> 0 Then 
    MsgBox errMsg, vbOKOnly, "警告" 
  End If 

End Sub 

表 6-3(2/2) イベントプロシージャ 

 

  

'｢物性値一覧表｣ボタンのクリック 
Private Sub ButMaterial_Click() 
  System.Cursor = wdCursorWait       'マウスポインタの設定 

  Application.ScreenUpdating = False '画面の更新を停止する 
   
  'Femap モデルファイルの MATERIALデータを表で出力 

  FeMaterial 
   
  Application.ScreenUpdating = True '画面を更新する 

  System.Cursor = wdCursorNormal    'マウスポインタの設定 
End Sub 
 

'｢板部材一覧表｣ボタンのクリック 
Private Sub ButPShell_Click() 
  System.Cursor = wdCursorWait       'マウスポインタの設定 

  Application.ScreenUpdating = False '画面の更新を停止する 
   
  'Femap モデルファイルの PROPATY(板)データを表で出力 

  FePropatyShell 
     
  Application.ScreenUpdating = True '画面を更新する 

  System.Cursor = wdCursorNormal    'マウスポインタの設定 
End Sub 
 

'｢梁部材一覧表｣ボタンのクリック 
Private Sub ButPBar_Click() 
  System.Cursor = wdCursorWait       'マウスポインタの設定 

  Application.ScreenUpdating = False '画面の更新を停止する 
   
  'Femap モデルファイルの PROPATY(梁)データを表で出力 

  FePropatyBar 
     
  Application.ScreenUpdating = True '画面を更新する 

  System.Cursor = wdCursorNormal    'マウスポインタの設定 
End Sub 
 

'｢質量一覧表｣ボタンのクリック 
Private Sub ButPMass_Click() 
  System.Cursor = wdCursorWait       'マウスポインタの設定 

  Application.ScreenUpdating = False '画面の更新を停止する 
   
  'Femap モデルファイルの PROPATY(質量)データを表で出力 

  FePropatyMass 
     
  Application.ScreenUpdating = True '画面を更新する 

  System.Cursor = wdCursorNormal    'マウスポインタの設定 
End Sub 
 

'｢固有振動数一覧表｣ボタンのクリック 
Private Sub ButCycle_Click() 
  System.Cursor = wdCursorWait       'マウスポインタの設定 

  Application.ScreenUpdating = False '画面の更新を停止する 
   
  'Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数シートに出力 

  FeCycle 
     
  Application.ScreenUpdating = True '画面を更新する 

  System.Cursor = wdCursorNormal    'マウスポインタの設定 
End Sub 

表 6-4(1/2) Femapモデルファイルの MATERIALデータの一覧表出力 

 

  

'=================================================================================== 
'Femapモデルファイルの MATERIAL データを表で出力 
'Flag : 全データの処理の場合は True 

'=================================================================================== 
Function FeMaterial(Optional Flag As Boolean = False) As String 
  Dim i As Long 

  Dim errMsg As String    'エラーメッセージ 
  Dim femap As Object   'Femap 
  Dim tbl As Table      'テーブル 

   
  Dim numRow As Integer '行数 
  Dim numCol As Integer '列数 

  Dim Row As Integer    '行カウント 
 
  Dim StrHead  As Variant 'ヘッダ 

  StrHead = Array("MID", "タイトル", "ヤング率" & vbCr & "(N/mm2)", "ポアソン比", "密度" & vbCr & "(N/mm3)") 
   
  On Error GoTo myError 

 
  'すでに実行されている FEMAP に接続 
  Set femap = GetObject(, "femap.model") 

  On Error GoTo 0 
   
  'Material オブジェクトを生成 

  Dim mt As Object 
  Set mt = femap.feMatl() 
 

  '物性値があるかチェック 
  If mt.First = 0 Then '物性値が無かった 
    If Flag Then 

      FeMaterial = "" 
    Else 
      MsgBox "物性値データはありませんでした。", vbOKOnly, "終了" 

    End If 
    Exit Function 
  End If 

   
  '----------------------------------------------------------- 
  'ワード出力 

  '----------------------------------------------------------- 
  If Not Flag Then 
    '新しい文書を開く 

    Documents.Add Template:="Normal", NewTemplate:=False, DocumentType:=wdNewBlankDocument 
    'ワードの初期設定 
    errMsg = InitWord 

    If Len(errMsg) <> 0 Then 
      MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 
      Exit Function 

    End If 
  End If 
   

  '文書に文字がある場合は改ページ 
  If ActiveDocument.Range.ComputeStatistics(wdStatisticWords) <> 0 Then 
    ActiveDocument.Bookmarks("\EndOfDoc").Select  '文末に移動 

    Selection.InsertBreak Type:=wdPageBreak       '改ページ 
  End If 
   

  Selection.TypeText Text:=TABLENAME_MAT '表タイトル 
  Selection.ParagraphFormat.Alignment = wdAlignParagraphCenter '中心揃え 
  Selection.TypeParagraph 

 
  numRow = mt.CountSet + 1        '表の行数、Material 数＋1 
  numCol = (UBound(StrHead)) + 1  '表の列数 

表 6-4(2/2) Femapモデルファイルの MATERIALデータの一覧表出力 

 
  

  'テーブルの作成 
  Set tbl = ActiveDocument.Tables.Add(Selection.Range, numRow, numCol, wdWord9TableBehavior) 
     

  '表に文字列を追加 
  With tbl 
    .Rows.Alignment = wdAlignRowCenter '表の位置をページ中央に設定 

         
    'ヘッダの設定 
    For i = 0 To numCol - 1 

      .Cell(1, i + 1).Range.Text = StrHead(i) 
      If i = 2 Or i = 4 Then 
        .Cell(1, i + 1).Range.Select 

         Call Superscript(2, 1, True) '上付き文字 
      End If 
    Next 

     
    'Femap モデルデータの Materialを表示 
    mt.First 

    Row = 2 
    Do 
      .Cell(Row, 1).Range.Text = mt.ID                        'ID 

      .Cell(Row, 2).Range.Text = mt.Title                     'タイトル 
      Call SetFormat(tbl, Row, 3, mt.Ex, 4)                   'ヤング率 
      .Cell(Row, 4).Range.Text = Format(mt.Nuxy, "0.0######") 'ポアソン比 

      Call SetFormat(tbl, Row, 5, mt.Density, 4)              '密度 
      Row = Row + 1 
    Loop While mt.Next 

     
    '表の列幅の設定 
    Dim ColWidth As Variant 

    ColWidth = Array(15, 50, 25, 25, 25) 
    Call SetWidth(tbl, ColWidth) 
               

    '表の列の文字位置の設定 
    Dim ColAlign As Variant '文字位置 
    Dim IndentL As Variant  'インデント左 

    Dim IndentR As Variant  'インデント右 
    ColAlign = Array(1, 0, 2, 0, 2) 
    IndentL = Array(-0.2, -0.2, -0.2, 2, -0.2) 

    IndentR = Array(-0.2, -0.2, 0, -0.2, 0) 
    Call SetAlignWidth(tbl, ColAlign, IndentL, IndentR) 
     

    '表ヘッダ書式の設定 
    Call SetHeader(tbl) 
  End With 

   
  ActiveDocument.Range(0, 0).Select '文書のトップに移動 
   

  FeMaterial = "" 
  Exit Function 
   

myError: 
  If Not Flag Then '単独処理の場合 
    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 

  Else 
    FeMaterial = "Femap モデルファイルを開いて下さい。" 
  End If 

End Function 

表 6-5(1/3) Femapモデルファイルの PROPATY(板)データの一覧表出力 

  

'=================================================================================== 
'Femapモデルファイルの PROPATY(板)データを表で出力 
'Flag : 全データの処理の場合は True 

'=================================================================================== 
Function FePropatyShell(Optional Flag As Boolean = False) As String 
  Dim i As Long 

  Dim errMsg As String    'エラーメッセージ 
  Dim femap As Object   'Femap 
  Dim tbl As Table      'テーブル 

 
  Dim numRow As Integer '行数 
  Dim numCol As Integer '列数 

  Dim Row As Integer    '行カウント 
 
  Dim StrHead  As Variant 'ヘッダ 

  StrHead = Array("PID", "タイトル", "MID", "板厚" & vbCr & "(mm)") 
   
  On Error GoTo myError 

  'すでに実行されている FEMAP に接続 
  Set femap = GetObject(, "femap.model") 
  On Error GoTo 0 

 
  'Propaty オブジェクトを生成 
  Dim pr As Object 

  Set pr = femap.feProp() 
   
  '板要素プロパティがあるかチェック 

  Dim PropFlag As Boolean 'プロパティ有無のフラッグ 
  pr.First 
  Do 

    If pr.Type = 17 Then 
      PropFlag = True 
      Exit Do 

    End If 
  Loop While pr.Next 
   

  If Not PropFlag Then 'プロパティーが無かった 
    If Flag Then 
      FePropatyShell = "" 

    Else 
      MsgBox "板要素プロパティはありませんでした。", vbOKOnly, "終了" 
    End If 

    Exit Function 
  End If 
   

  '----------------------------------------------------------- 
  'ワード出力 
  '----------------------------------------------------------- 

  If Not Flag Then 
    '新しい文書を開く 
    Documents.Add Template:="Normal", NewTemplate:=False, DocumentType:=wdNewBlankDocument 

    'ワードの初期設定 
    errMsg = InitWord 
    If Len(errMsg) <> 0 Then 

      MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 
      Exit Function 
    End If 

  End If 
     
  '文書に文字がある場合は改ページ 

  If ActiveDocument.Range.ComputeStatistics(wdStatisticWords) <> 0 Then 
    ActiveDocument.Bookmarks("\EndOfDoc").Select  '文末に移動 
    Selection.InsertBreak Type:=wdPageBreak       '改ページ 

  End If 

表 6-5(2/3) Femapモデルファイルの PROPATY(板)データの一覧表出力 

  

  Selection.TypeText Text:=TABLENAME_PSHELL '表タイトル 
  Selection.ParagraphFormat.Alignment = wdAlignParagraphCenter '中心揃え 

  Selection.TypeParagraph 
   
  '板 Propaty 数のカウント 

  pr.First 
  Do 
    If pr.Type = 17 Then '板の場合 

      numRow = numRow + 1 
    End If 
  Loop While pr.Next 

   
  numRow = numRow + 1             '表の行数、板 Propaty 数＋1 
  numCol = (UBound(StrHead)) + 1  '表の列数 

   
  'テーブルの作成 
  Set tbl = ActiveDocument.Tables.Add(Selection.Range, numRow, numCol, wdWord9TableBehavior) 

   
  '表に文字列を追加 
  With tbl 

    .Rows.Alignment = wdAlignRowCenter '表の位置をページ中央に設定 
     
    'ヘッダの設定 

    For i = 0 To numCol - 1 
      .Cell(1, i + 1).Range.Text = StrHead(i) 
    Next 

     
    'Femap モデルデータの Propaty(板)を表示 
    pr.First 

    Row = 2 
    Do 
      If pr.Type = 17 Then '板の場合 

        .Cell(Row, 1).Range.Text = pr.ID       'ID 
        .Cell(Row, 2).Range.Text = pr.Title    'タイトル 
        .Cell(Row, 3).Range.Text = pr.matlID   'MID 

        .Cell(Row, 4).Range.Text = Format(pr.pval(0), "0.0") '板厚 
        Row = Row + 1 
      End If 

    Loop While pr.Next 
     
    '表の列幅の設定 

    Dim ColWidth As Variant 
    ColWidth = Array(15, 50, 15, 23) 
    Call SetWidth(tbl, ColWidth) 

               
    '表の列の文字位置の設定 
    Dim ColAlign As Variant '文字位置 

    Dim IndentL As Variant  'インデント左 
    Dim IndentR As Variant  'インデント右 
    ColAlign = Array(1, 0, 1, 2) 

    IndentL = Array(-0.2, -0.2, -0.2, -0.2) 
    IndentR = Array(-0.2, -0.2, -0.2, 1.5) 
    Call SetAlignWidth(tbl, ColAlign, IndentL, IndentR) 

     
    '表ヘッダ書式の設定 
    Call SetHeader(tbl) 

  End With 

表 6-5(3/3) Femapモデルファイルの PROPATY(板)データの一覧表出力 

  

   ActiveDocument.Range(0, 0).Select '文書のトップに移動 
   

  FePropatyShell = "" 
  Exit Function 
   

myError: 
  If Not Flag Then '単独処理の場合 
    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 

  Else 
    FePropatyShell = "Femap モデルファイルを開いて下さい。" 
  End If 

End Function 

表 6-6(1/3) Femapモデルファイルの PROPATY(梁)データの一覧表出力 

 

  

'=================================================================================== 
'Femapモデルファイルの PROPATY(梁)データを表で出力 
'Flag : 全データの処理の場合は True 

'=================================================================================== 
Function FePropatyBar(Optional Flag As Boolean = False) As String 
  Dim i As Long 

  Dim errMsg As String    'エラーメッセージ 
  Dim femap As Object   'Femap 
  Dim tbl As Table      'テーブル 

 
  Dim numRow As Integer '行数 
  Dim numCol As Integer '列数 

  Dim Row As Integer    '行カウント 
 
  Dim StrHead  As Variant 'ヘッダ 

  StrHead = Array("PID", "タイトル", "MID", "断面積" & vbCr & "A(mm2)", "断面二次" & vbCr & "IY(mm4)", "断面二
次" & vbCr & "IZ(mm4)", "ねじり" & vbCr & "J(mm4)") 
   

  On Error GoTo myError 
   
  'すでに実行されている FEMAP に接続 

  Set femap = GetObject(, "femap.model") 
  On Error GoTo 0 
 

  'Propaty オブジェクトを生成 
  Dim pr As Object 
  Set pr = femap.feProp() 

   
  '梁要素プロパティがあるかチェック 
  Dim PropFlag As Boolean 'プロパティ有無のフラッグ 

  pr.First 
  Do 
    If pr.Type = 2 Then  'バー 

      If pr.flagI(1) = 0 Or pr.flagI(1) = 1 Or pr.flagI(1) = 5 Or pr.flagI(1) = 6 Then   '形状入力は処理しない 
        PropFlag = True 
        Exit Do 

      End If 
    End If 
  Loop While pr.Next 

   
  If Not PropFlag Then 'プロパティーが無かった 
    If Flag Then 

      FePropatyBar = "" 
    Else 
      MsgBox "梁要素プロパティはありませんでした。", vbOKOnly, "終了" 

    End If 
    Exit Function 
  End If 

   
  '----------------------------------------------------------- 
  'ワード出力 

  '----------------------------------------------------------- 
  If Not Flag Then 
    '新しい文書を開く 

    Documents.Add Template:="Normal", NewTemplate:=False, DocumentType:=wdNewBlankDocument 
    'ワードの初期設定 
    errMsg = InitWord 

    If Len(errMsg) <> 0 Then 
      MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 
      Exit Function 

    End If 
  End If 

表 6-6(2/3) Femapモデルファイルの PROPATY(梁)データの一覧表出力 

 
  

       
  '文書に文字がある場合は改ページ 
  If ActiveDocument.Range.ComputeStatistics(wdStatisticWords) <> 0 Then 

    ActiveDocument.Bookmarks("\EndOfDoc").Select  '文末に移動 
    Selection.InsertBreak Type:=wdPageBreak       '改ページ 
  End If 

   
  Selection.TypeText Text:=TABLENAME_PBAR '表タイトル 
  Selection.ParagraphFormat.Alignment = wdAlignParagraphCenter '中心揃え 

  Selection.TypeParagraph 
   
  'バーPropaty 数のカウント 

  pr.First 
  Do 
    If pr.Type = 2 Then 'バー 

      If pr.flagI(1) = 0 Or pr.flagI(1) = 1 Or pr.flagI(1) = 5 Or pr.flagI(1) = 6 Then   '形状入力は処理しない 
        numRow = numRow + 1 
      End If 

    End If 
  Loop While pr.Next 
   

  numRow = numRow + 1             '表の行数、バーPropaty 数＋1 
  numCol = (UBound(StrHead)) + 1  '表の列数 
   

  'テーブルの作成 
  Set tbl = ActiveDocument.Tables.Add(Selection.Range, numRow, numCol, wdWord9TableBehavior) 
   

  '表に文字列を追加 
  With tbl 
    .Rows.Alignment = wdAlignRowCenter '表の位置をページ中央に設定 

    .Rows(1).Select 
    Selection.Cells.VerticalAlignment = wdCellAlignVerticalCenter '鉛直 中央 
     

    'ヘッダの設定 
    For i = 0 To numCol - 1 
      .Cell(1, i + 1).Range.Text = StrHead(i) 

      If i >= 3 Then 
        .Cell(1, i + 1).Range.Select 
         Call Superscript(2, 1, True) '上付き文字 

      End If 
    Next 
     

    'Femap モデルデータの Propaty(梁)を表示 
    pr.First 
    Row = 2 

    Do 
      If pr.Type = 2 Then 'バー 
        If pr.flagI(1) = 0 Or pr.flagI(1) = 1 Or pr.flagI(1) = 5 Or pr.flagI(1) = 6 Then   '形状入力は処理しな

い 
          .Cell(Row, 1).Range.Text = pr.ID        'ID 
          .Cell(Row, 2).Range.Text = pr.Title     'タイトル 

          .Cell(Row, 3).Range.Text = pr.matlID    'MID 
          Call SetFormat(tbl, Row, 4, pr.pval(0), 4) 'A 
          Call SetFormat(tbl, Row, 5, pr.pval(2), 4) 'Iy 

          Call SetFormat(tbl, Row, 6, pr.pval(1), 4) 'Iz 
          Call SetFormat(tbl, Row, 7, pr.pval(4), 4) 'J 
          Row = Row + 1 

        End If 
      End If 
    Loop While pr.Next 

表 6-6(3/3) Femapモデルファイルの PROPATY(梁)データの一覧表出力 

 

  

    '表の列幅の設定 

    Dim ColWidth As Variant 
    ColWidth = Array(15, 50, 15, 25, 25, 25, 25) 
    Call SetWidth(tbl, ColWidth) 

           
    '表の列の文字位置の設定 
    Dim ColAlign As Variant '文字位置 

    Dim IndentL As Variant  'インデント左 
    Dim IndentR As Variant  'インデント右 
    ColAlign = Array(1, 0, 2, 2, 2, 2, 2) 

    IndentL = Array(-0.2, -0.2, -0.2, -0.2, -0.2, -0.2, -0.2) 
    IndentR = Array(-0.2, -0.2, -0.2, -0.2, -0.2, -0.2, -0.2) 
    Call SetAlignWidth(tbl, ColAlign, IndentL, IndentR) 

     
    '表ヘッダ書式の設定 
    Call SetHeader(tbl) 

  End With 
   
  ActiveDocument.Range(0, 0).Select '文書のトップに移動 

   
  FePropatyBar = "" 
  Exit Function 

   
myError: 
  If Not Flag Then '単独処理の場合 

    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 
  Else 
    FePropatyBar = "Femap モデルファイルを開いて下さい。" 

  End If 
End Function 



ZW-FmpGetMdl Ver.0.1 Femap モデルファイルの読込み例 Word マクロ使用説明書 

 

－21－ 

表 6-7(1/3) Femapモデルファイルの PROPATY(質量)データの一覧表出力 

 

  

'=================================================================================== 
'Femapモデルファイルの PROPATY(質量)データを表で出力 
'Flag : 全データの処理の場合は True 

'=================================================================================== 
Function FePropatyMass(Optional Flag As Boolean = False) As String 
  Dim i As Long 

  Dim errMsg As String    'エラーメッセージ 
  Dim femap As Object   'Femap 
  Dim tbl As Table      'テーブル 

   
  Dim numRow As Integer '行数 
  Dim numCol As Integer '列数 

  Dim Row As Integer    '行カウント 
 
  Dim StrHead  As Variant 'ヘッダ 

  StrHead = Array("PID", "タイトル", "質量" & vbCr & "(N/mm3)", "オフセット X" & vbCr & "(mm)", "オフセット Y" 
& vbCr & "(mm)", "オフセット Z" & vbCr & "(mm)") 
   

  On Error GoTo myError 
   
  'すでに実行されている FEMAP に接続 

  Set femap = GetObject(, "femap.model") 
  On Error GoTo 0 
 

  'Propaty オブジェクトを生成 
  Dim pr As Object 
  Set pr = femap.feProp() 

   
  '質量要素プロパティがあるかチェック 
  Dim PropFlag As Boolean 'プロパティ有無のフラッグ 

  pr.First 
  Do 
    If pr.Type = 27 Then '質量 

      PropFlag = True 
      Exit Do 
    End If 

  Loop While pr.Next 
   
  If Not PropFlag Then 'プロパティーが無かった 

    If Flag Then 
      FePropatyMass = "" 
    Else 

      MsgBox "質量要素プロパティはありませんでした。", vbOKOnly, "終了" 
    End If 
    Exit Function 

  End If 
   
  '----------------------------------------------------------- 

  'ワード出力 
  '----------------------------------------------------------- 
  If Not Flag Then 

    '新しい文書を開く 
    Documents.Add Template:="Normal", NewTemplate:=False, DocumentType:=wdNewBlankDocument 
    'ワードの初期設定 

    errMsg = InitWord 
    If Len(errMsg) <> 0 Then 
      MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 

      Exit Function 
    End If 
  End If 

表 6-7(2/3) Femapモデルファイルの PROPATY(質量)データの一覧表出力 

 

  

       

  '文書に文字がある場合は改ページ 
  If ActiveDocument.Range.ComputeStatistics(wdStatisticWords) <> 0 Then 
    ActiveDocument.Bookmarks("\EndOfDoc").Select  '文末に移動 

    Selection.InsertBreak Type:=wdPageBreak       '改ページ 
  End If 
   

  Selection.TypeText Text:=TABLENAME_PMASS '表タイトル 
  Selection.ParagraphFormat.Alignment = wdAlignParagraphCenter '中心揃え 
  Selection.TypeParagraph 

   
  '質量 Propaty 数のカウント 
  pr.First 

  Do 
    If pr.Type = 27 Then '質量 
      numRow = numRow + 1 

    End If 
  Loop While pr.Next 
   

  numRow = numRow + 1             '表の行数、質量 Propaty 数＋1 
  numCol = (UBound(StrHead)) + 1  '表の列数 
   

  'テーブルの作成 
  Set tbl = ActiveDocument.Tables.Add(Selection.Range, numRow, numCol, wdWord9TableBehavior) 
   

  '表に文字列を追加 
  With tbl 
    .Rows.Alignment = wdAlignRowCenter '表の位置をページ中央に設定 

    .Rows(1).Select 
    Selection.Cells.VerticalAlignment = wdCellAlignVerticalCenter '鉛直 中央 
     

    'ヘッダの設定 
    For i = 0 To numCol - 1 
      .Cell(1, i + 1).Range.Text = StrHead(i) 

      If i = 2 Then 
        .Cell(1, i + 1).Range.Select 
         Call Superscript(2, 1, True) '上付き文字 

      End If 
    Next 
     

    'Femap モデルデータの Propaty(質量)を表示 
    pr.First 
    Row = 2 

    Do 
      If pr.Type = 27 Then '質量 
        .Cell(Row, 1).Range.Text = pr.ID                      'ID 

        .Cell(Row, 2).Range.Text = pr.Title                   'タイトル 
        .Cell(Row, 3).Range.Text = StrFit(pr.pval(7), 4)      '質量 
        .Cell(Row, 4).Range.Text = Format(pr.pval(8), "0.0")  'オフセット X 

        .Cell(Row, 5).Range.Text = Format(pr.pval(9), "0.0")  'オフセット Y 
        .Cell(Row, 6).Range.Text = Format(pr.pval(10), "0.0") 'オフセット Z 
        Row = Row + 1 

      End If 
    Loop While pr.Next 

表 6-7(3/3) Femapモデルファイルの PROPATY(質量)データの一覧表出力 

 

  

    '表の列幅の設定 
    Dim ColWidth As Variant 

    ColWidth = Array(15, 50, 25, 25, 25, 25) 
    Call SetWidth(tbl, ColWidth) 
       

    '表の列の文字位置の設定 
    Dim ColAlign As Variant '文字位置 
    Dim IndentL As Variant  'インデント左 

    Dim IndentR As Variant  'インデント右 
    ColAlign = Array(1, 0, 0, 2, 2, 2) 
    IndentL = Array(-0.2, -0.2, 0.5, -0.2, -0.2, -0.2, -0.2) 

    IndentR = Array(-0.2, -0.2, -0.2, 2, 2, 2) 
    Call SetAlignWidth(tbl, ColAlign, IndentL, IndentR) 
     

    '表ヘッダ書式の設定 
    Call SetHeader(tbl) 
  End With 

   
  ActiveDocument.Range(0, 0).Select '文書のトップに移動 
   

  FePropatyMass = "" 
  Exit Function 
   

myError: 
  If Not Flag Then '単独処理の場合 
    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 

  Else 
    FePropatyMass = "Femap モデルファイルを開いて下さい。" 
  End If 

End Function 

表 6-8(1/2) Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数の一覧表出力 

  

'=================================================================================== 
'Femapモデルファイルの OutputSet データから固有値計算結果の次数と周波数を表で出力 
'Flag : 全データの処理の場合は True 

'=================================================================================== 
Function FeCycle(Optional Flag As Boolean = False) As String 
  Dim i As Long 

  Dim errMsg As String    'エラーメッセージ 
  Dim femap As Object   'Femap 
  Dim tbl As Table      'テーブル 

 
  Dim numRow As Integer '行数 
  Dim numCol As Integer '列数 

  Dim Row As Integer    '行カウント 
  Dim Mode As Integer   '次数 
 

  Dim StrHead  As Variant 'ヘッダ 
  StrHead = Array("次数", "固有振動数" & vbCr & "(Hz)") 
   

  Const MAX_OUTPUT = 30 
   
  On Error GoTo myError 

   
  'すでに実行されている FEMAP に接続 
  Set femap = GetObject(, "femap.model") 

  On Error GoTo 0 
 
  'OutputSet オブジェクトを生成 

  Dim Os As Object 
  Set Os = femap.feOutputSet 
   

  '固有値計算結果があるかチェック 
  If Os.First = 0 Then '計算結果が無かった 
    If Flag Then 

      FeCycle = "" 
    Else 
      MsgBox "固有値計算結果はありませんでした。", vbOKOnly, "終了" 

    End If 
    Exit Function 
  End If 

   
  '----------------------------------------------------------- 
  'ワード出力 

  '----------------------------------------------------------- 
  If Not Flag Then 
    '新しい文書を開く 

    Documents.Add Template:="Normal", NewTemplate:=False, DocumentType:=wdNewBlankDocument 
    'ワードの初期設定 
    errMsg = InitWord 

    If Len(errMsg) <> 0 Then 
      MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 
      Exit Function 

    End If 
  End If 
       

  '文書に文字がある場合は改ページ 
  If ActiveDocument.Range.ComputeStatistics(wdStatisticWords) <> 0 Then 
    ActiveDocument.Bookmarks("\EndOfDoc").Select  '文末に移動 

    Selection.InsertBreak Type:=wdPageBreak       '改ページ 
  End If 
  Selection.TypeText Text:=TABLENAME_CYCLE '表タイトル 

  Selection.ParagraphFormat.Alignment = wdAlignParagraphCenter '中心揃え 
  Selection.TypeParagraph 

表 6-8(2/2) Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数の一覧表出力 

  

  numRow = Os.CountSet + 1        '表の行数、OutputSet数＋1 
  numCol = (UBound(StrHead)) + 1  '表の列数 
  If numRow > MAX_OUTPUT + 1 Then '次数が MAX_OUTPUT を超える場合は MAX_OUTPUT+1 を行数とする 

    numRow = MAX_OUTPUT + 1 
  End If 
   

  'テーブルの作成 
  Set tbl = ActiveDocument.Tables.Add(Selection.Range, numRow, numCol, wdWord9TableBehavior) 
   

  '表に文字列を追加 
  With tbl 
    .Rows.Alignment = wdAlignRowCenter '表の位置をページ中央に設定 

    .Rows(1).Select 
    Selection.Cells.VerticalAlignment = wdCellAlignVerticalCenter '鉛直 中央 
     

    'ヘッダの設定 
    For i = 0 To numCol - 1 
      .Cell(1, i + 1).Range.Text = StrHead(i) 

      If i = 2 Then 
        .Cell(1, i + 1).Range.Select 
         Call Superscript(2, 1, True) '上付き文字 

      End If 
    Next 
     

    'Femap モデルデータの周波数を表示 
    Os.First 
    Row = 2 

    Do 
      Mode = Mode + 1 
      .Cell(Row, 1).Range.Text = Mode      '次数 

      .Cell(Row, 2).Range.Text = Format(Os.Value, "0.000")  '周波数 
      If Mode = MAX_OUTPUT Then Exit Do    '次数が MAX_OUTPUT を超える場合は MAX_OUTPUT で終了 
      Row = Row + 1 

    Loop While Os.Next 
   
   '表の列幅の設定 

    Dim ColWidth As Variant 
    ColWidth = Array(15, 23) 
    Call SetWidth(tbl, ColWidth) 

     
    '表の列の文字位置の設定 
    Dim ColAlign As Variant '文字位置 

    Dim IndentL As Variant  'インデント左 
    Dim IndentR As Variant  'インデント右 
    ColAlign = Array(1, 2) 

    IndentL = Array(-0.2, 1) 
    IndentR = Array(-0.2, 1) 
    Call SetAlignWidth(tbl, ColAlign, IndentL, IndentR) 

     
    '表ヘッダ書式の設定 
    Call SetHeader(tbl) 

  End With 
   
  ActiveDocument.Range(0, 0).Select '文書のトップに移動 

   
  FeCycle = "" 
  Exit Function 

   
myError: 
  If Not Flag Then '単独処理の場合 

    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告" 
  Else 
    FeCycle = "Femap モデルファイルを開いて下さい。" 

  End If 
End Function 

表 6-9 ワードの初期設定 

 

  

'================================================================================ 
'ワードの初期設定 

'================================================================================ 
Function InitWord() As String 
  Dim errMsg As String 

     
  Selection.EndKey Unit:=wdStory '文書の最後に移動 
   

  ActiveWindow.ActivePane.View.Zoom.Percentage = 100 'ワード表示の% 
   
  'ページ設定 

  errMsg = InitDocPage 
  If Len(errMsg) <> 0 Then 
    InitWord = errMsg 

    Exit Function 
  End If 
   

  'フォント初期設 
  errMsg = InitDocFont 
  If Len(errMsg) <> 0 Then 

    InitWord = errMsg 
    Exit Function 
  End If 

   
  '書式の初期設定 
  errMsg = InitDocParagraphFormat 

  If Len(errMsg) <> 0 Then 
    InitWord = errMsg 
    Exit Function 

  End If 
   
  InitWord = "" 

End Function 

表 6-10 ページ設定 

 

  

'================================================================================ 

'ページ設定 
'================================================================================ 
Function InitDocPage() As String 

If False Then 
  Selection.WholeStory 
  With ActiveDocument.Styles(wdStyleNormal).Font 

    If .NameFarEast = .NameAscii Then 
      .NameAscii = "" 
    End If 

    .NameFarEast = "" 
  End With 
End If 

  With ActiveDocument.PageSetup 
    .LineNumbering.Active = False 
    .Orientation = wdOrientPortrait 

    .TopMargin = MillimetersToPoints(31) 
    .BottomMargin = MillimetersToPoints(25) 
    .LeftMargin = MillimetersToPoints(25) 

    .RightMargin = MillimetersToPoints(25) 
    .Gutter = MillimetersToPoints(0) 
    .HeaderDistance = MillimetersToPoints(15) 

    .FooterDistance = MillimetersToPoints(17.5) 
    .PageWidth = MillimetersToPoints(210) 
    .PageHeight = MillimetersToPoints(297) 

    .FirstPageTray = wdPrinterDefaultBin 
    .OtherPagesTray = wdPrinterDefaultBin 
    .SectionStart = wdSectionNewPage 

    .OddAndEvenPagesHeaderFooter = False 
    .DifferentFirstPageHeaderFooter = False 
    .VerticalAlignment = wdAlignVerticalTop 

    .SuppressEndnotes = False 
    .MirrorMargins = False 
    .TwoPagesOnOne = False 

    .BookFoldPrinting = False 
    .BookFoldRevPrinting = False 
    .BookFoldPrintingSheets = 1 

    .GutterPos = wdGutterPosLeft 
    .LayoutMode = wdLayoutModeLineGrid 
  End With 

     
  InitDocPage = "" 
End Function 

表 6-11 フォント初期設 

 

  

'================================================================================ 

'フォント初期設 
'================================================================================ 
Function InitDocFont() As String 

  With Selection.Font 
    .NameFarEast = "ＭＳ 明朝" 
    .NameAscii = "ＭＳ 明朝" 

    .NameOther = "ＭＳ 明朝" 
    .Name = "ＭＳ 明朝" 
    .Size = 10.5 

    .Bold = False 
    .Italic = False 
    .Underline = wdUnderlineNone 

    .UnderlineColor = wdColorAutomatic 
    .StrikeThrough = False 
    .DoubleStrikeThrough = False 

    .Outline = False 
    .Emboss = False 
    .Shadow = False 

    .Hidden = False 
    .SmallCaps = False 
    .AllCaps = False 

    .Color = wdColorAutomatic 
    .Engrave = False 
    .Superscript = False 

    .Subscript = False 
    .Spacing = 0 
    .Scaling = 100 

    .Position = 0 
    .Kerning = 1 
    .Animation = wdAnimationNone 

    .DisableCharacterSpaceGrid = False 
    .EmphasisMark = wdEmphasisMarkNone 
    .Ligatures = wdLigaturesNone 

    .NumberSpacing = wdNumberSpacingDefault 
    .NumberForm = wdNumberFormDefault 
    .StylisticSet = wdStylisticSetDefault 

    .ContextualAlternates = 0 
  End With 
   

  InitDocFont = "" 
End Function 

表 6-12 書式の初期設定 

 
 

表 6-13 表の列幅の設定 

 

  

'================================================================================ 
'書式の初期設定 
'================================================================================ 

Function InitDocParagraphFormat() As String 
  With Selection.ParagraphFormat 
    .LeftIndent = MillimetersToPoints(0) 

    .RightIndent = MillimetersToPoints(0) 
    .SpaceBefore = 0 
    .SpaceBeforeAuto = False 

    .SpaceAfter = 0 
    .SpaceAfterAuto = False 
    .LineSpacingRule = wdLineSpaceSingle 

    .Alignment = wdAlignParagraphLeft 
    .WidowControl = False 
    .KeepWithNext = False 

    .KeepTogether = False 
    .PageBreakBefore = False 
    .NoLineNumber = False 

    .Hyphenation = True 
    .FirstLineIndent = MillimetersToPoints(0) 
    .OutlineLevel = wdOutlineLevelBodyText 

    .CharacterUnitLeftIndent = 0 
    .CharacterUnitRightIndent = 0 
    .CharacterUnitFirstLineIndent = 0 

    .LineUnitBefore = 0 
    .LineUnitAfter = 0 
    .MirrorIndents = False 

    .TextboxTightWrap = wdTightNone 
    .CollapsedByDefault = False 
    .AutoAdjustRightIndent = True 

    .DisableLineHeightGrid = False 
    .FarEastLineBreakControl = True 
    .WordWrap = True 

    .HangingPunctuation = True 
    .HalfWidthPunctuationOnTopOfLine = False 
    .AddSpaceBetweenFarEastAndAlpha = True 

    .AddSpaceBetweenFarEastAndDigit = True 
    .BaseLineAlignment = wdBaselineAlignAuto 
  End With 

   
  InitDocParagraphFormat = "" 
End Function 

'=================================================================================== 
'表の列幅の設定 

'=================================================================================== 
Sub SetWidth(tbl, ColWidth) 
  Dim i As Long 

   
  For i = 0 To UBound(ColWidth) 
    tbl.Columns(i + 1).Select 

    Selection.Columns.PreferredWidthType = wdPreferredWidthPoints 
    Selection.Columns.PreferredWidth = MillimetersToPoints(ColWidth(i)) 
  Next 

End Sub 

表 6-14 表ヘッダ書式の設定 

 

 

表 6-15 表の列の文字位置の設定 

 
 

表 6-16 上付き、下付き文字 

 

  

'=================================================================================== 

'表ヘッダ書式の設定 
'tbl:テーブル 
'=================================================================================== 

Sub SetHeader(tbl) 
  tbl.Rows(1).Select 
  With Selection.ParagraphFormat 

    Selection.Cells.VerticalAlignment = wdCellAlignVerticalCenter '鉛直 中央 
    .Alignment = wdAlignParagraphCenter   '中心揃え 
    .CharacterUnitLeftIndent = -0.2       '左 

    .CharacterUnitRightIndent = -0.2      '右 
    .LineSpacingRule = wdLineSpaceExactly '行間 固定 
    .LineSpacing = 12                     '行間隔 

    .SpaceBefore = 0                      '段落 前の行との間隔(ポイント) 
  End With 
End Sub 

'=================================================================================== 
'表の列の文字位置の設定 
'tbl:テーブル、ColAlign:文字位置 0=左詰、1=中央、2=右詰, IndentL:インデント左, IndentR:インデント右 

'=================================================================================== 
Sub SetAlignWidth(tbl, ColAlign, IndentL, IndentR) 
  Dim i As Long 

   
  With Selection.ParagraphFormat 
    For i = 0 To UBound(ColAlign) 

      tbl.Columns(i + 1).Select 
       
      Select Case ColAlign(i) 

        Case 0 
          .Alignment = wdAlignParagraphLeft   '左詰め 
        Case 1 

          .Alignment = wdAlignParagraphCenter '中央 
        Case 2 
          .Alignment = wdAlignParagraphRight  '右詰め 

      End Select 
      .CharacterUnitLeftIndent = IndentL(i)   'インデント 左文字数 
      .CharacterUnitRightIndent = IndentR(i)  'インデント 右左文字数 

    Next 
  End With 
End Sub 

 

'=================================================================================== 

'上付き、下付き文字 
'nLeft:最後の文字から左への移動文字数、nChar:修正する文字数、FlagTopBottom:True=上付き、False=下付き 
'=================================================================================== 

Sub Superscript(nLeft, nChar, FlagTopBottom) 
  Selection.EndKey Unit:=wdLine 
  Selection.MoveLeft Unit:=wdCharacter, Count:=nLeft 

  Selection.MoveRight Unit:=wdCharacter, Count:=nChar, Extend:=wdExtend 
  If FlagTopBottom Then 
    Selection.Font.Superscript = True '上付き 

  Else 
    Selection.Font.Subscript = True   '下付き 
  End If 

End Sub 

表 6-17 数値を文字列に変換し小数部が合うように文字左にブランクを追加 

 

 

表 6-18 指数値を文字列に変換して(1.234E+02→1.234×102、1.234E-02→1.234×10-2) 

指数を上付きに設定 

 

  

'=================================================================================== 

'数値を文字列に変換し小数部が合うように文字左にブランクを追加 
'val:数値、整数部の最大桁数 
'=================================================================================== 

Function StrFit(val, nDpart) 
  Dim Str As String   '出力文字 
  Dim fmt As String   '出力書式 

  Dim nChr As Integer '指数部の文字数 
   
  '数値を文字列に変換 

  fmt = "0." & String(nDpart, "#") 
  Str = Format(val, fmt) 
  StrFit = Space(nDpart - InStr(1, Str, ".")) & Str 

End Function 

'=================================================================================== 
'指数値を文字列に変換して(1.234E+02→1.234×102、1.234E-02→1.234×10-2)指数を上付きに設定 

'tbl:テーブル、Row:表の行インデックス、Col:表の列インデックス、val:数値、nDpart:少数部桁数 
'=================================================================================== 
Sub SetFormat(tbl, Row, Col, val, nDpart) 

  Dim Str As String   '出力文字 
  Dim fmt As String   '出力書式 
  Dim nChr As Integer '指数部の文字数 

  Dim SetFStrormat As String '文字列 
   
  '数値を文字列に変換 

  fmt = "0." & String(nDpart, "0") & "E+00" 
  Str = Format(val, fmt) 
   

  '指数部の文字を変換 
  If Log10(val) >= 10 Then 
    Str = Replace(Left(Str, Len(Str) - 2), "E+", "×10") & Right(Str, 2) 

    nChr = 2 
  ElseIf Log10(val) >= 1 And Log10(val) < 10 Then 
    Str = Replace(Left(Str, Len(Str) - 2), "E+", "×10") & Right(Str, 1) 

    nChr = 1 
  ElseIf Log10(val) >= 0 And Log10(val) < 1 Then 
    Str = Left(Str, Len(Str) - 4) 

    nChr = 0 
  ElseIf Log10(val) < 0 And Log10(val) >= -9 Then 
    Str = Replace(Left(Str, Len(Str) - 2), "E-", "×10-") & Right(Str, 1) 

    nChr = 2 
  ElseIf Log10(val) >= -90 Then 
    SetFStrormat = Replace(Left(Str, Len(Str) - 2), "E-", "×10-") & Right(Str, 2) 

    nChr = 3 
  Else 
    Exit Sub 

  End If 
   
  '文字をセルに設定し、指数部のフォントを上付きに変更 

  With tbl 
    .Cell(Row, Col).Range.Text = Str 
    .Cell(Row, Col).Range.Select 

    Call Superscript(nChr, nChr, True) '上付文字 
  End With 
End Sub 

表 6-19 log10 

 

'=================================================================================== 

'log10 
'=================================================================================== 
Function Log10(X) 

  Log10 = Log(Abs(X)) / Log(10#) 
End Function 


